DESIGNED FOR HP-IB SYSTEMS

Programming Note

8566A/8568A/9826/9836-1 **NOVEMBER 1982**

SUPERCEDES: NONE

Introductory Operating Guide
for the 8566A/8568A Spectrum Analyzers with the 9826/9836 Desktop Computers



HEWLETT PACKARD

Introduction

This note is an introductory guide to remote operation and programming of 8566A and 8568A Spectrum Analyzers using either the 9826A or 9836A Desktop Computer with BASIC. Included in this guide are system connections for remote operation and several example programs with descriptions of each step.

The 9826A and 9836A use HPL and Pascal as well as BASIC. Although this guide is based on BASIC, the setup and language independent programming techniques apply equally well to systems using HPL or Pascal.

The 8566A and 8568A are microprocessor-controlled, general-purpose spectrum analyzers which are compatible with the Hewlett-Packard Interface Bus (HP-IB). When used with any HP-IB controller, such as the 9826A or 9836A, these instruments become fully automated spectrum analyzers featuring:

- Precise, stable LO tuning
- High sensitivity and resolution
- Wide dynamic range

Related Documents

Complete operating information for the 8566A/8568A analyzers can be found in:

1. 8566A/8568A Spectrum Analyzer Operation (P/N 08566-90002 or 08568-90002)
2. 8566A/8568A Spectrum Analyzer Remote Operation (P/N 08566-90003 or 08568-90003)
3. 8566A/8568A Spectrum Analyzer Pull-Out Information Cards

Information on operating the 9826A and 9836A controllers can be found in:

1. Basic Operating Manual (P/N 09826-90000)
2. Basic Programming Techniques (P/N 09826-90010)
3. Basic Interfacing Techniques (P/N 09826-90020)
4. Basic Language Reference (P/N 09826-90055)

Equipment Required

To perform the examples in this note, you will need the following equipment and accessories:

1. 8566A or 8568A Spectrum Analyzer
2. 9826A or 9836A Desktop Computer with ROM-based or RAM-based BASIC language (Options 011 or 711)
3. 10833 A/B/C/D HP-IB Cable

Setup

Figure 1 shows the system connections. To connect the system as shown, follow these steps.

1. Turn off the power to the 9826A/9836A and 8566A/8568A.
2. Attach an HP-IB cable to the 24-pin HP-IB connectors on the back panels of the 9826A/9836A and 8566A/8568A. The connectors are shaped to ensure proper orientation. (See Figure 1.)

CAUTION

Do not attempt to mate silver English threaded screws on one connector with black metric threaded nuts on another connector, or vice versa, as damage to the hardware may result. A metric conversion kit which will convert

one cable and one or two instruments to metric hardware may be obtained by ordering HP P/N 5060-0138.

Check-Out

Determine whether your 9826A/9836A has a soft-loaded (RAM) or built-in (ROM) language system and follow the appropriate procedure below:

Built-in System

1. Remove any disc in the drive and then press the power switch in.
2. A "BASIC READY" message should appear and the computer is now ready for use.
3. If more than one language system is built-in, BASIC (B) and HPL (H) for example, the computer will display:

   WHICH SYSTEM?
   B H

   Press the "B" key to select the BASIC system.

Soft-loaded System

1. Insert the BASIC language system disc into the disc drive and close the door.
2. Press the power switch in.
3. After a few seconds the "BASIC READY" message should appear and the computer is now ready for use.

If the computer does not display the "READY" message after the procedures above are completed, refer to the BASIC Operating Manual, Chapter 1.

After making AC power line connections to the analyzer, the STANDBY lights on both the RF and display sections should be illuminated. Set the analyzer to LINE ON.



Figure 1. System Connection



LINE ON and STANDBY



INSTR CHECK LED

Upon LINE ON, the analyzer will perform an automatic internal instrument check, designated by the red INSTR CHECK indicators. Both LED's will turn on momentarily during the brief check routine and, if the instrument is operating properly, will go off during operation, except when another instrument check is triggered by an Instrument Preset. If one or both LED's remain on, refer to the 8566A/8568A Operating and Service Manual, Section II.

Verify that the analyzer's address is set to 18. The read/write address of the 8566A or 8568A can be determined and altered from the front panel by using the shift function P:

Pressing [SHIFT] [ZOOM] (P) [1] [8] [Hz µV µsec]

sets the address to 18.

When the analyzer is turned on from a cold state, CRT messages OVEN COLD and REF UNLOCK may appear. These will go off typically ten minutes after AC power is connected. Type the following commands on the controller keyboard.

```
ABORT 7      (Press EXECUTE)
REMOTE 718   (Press EXECUTE)
```

If ADRS'D and REM light up on the analyzer's front panel, proceed to the programming examples. If either ADRS'D or REM do not light, check to make sure that the interface cables are properly connected and the address in the REMOTE statement matches the address of the 8566A/8568A. Although 18 is the factory- set address and the address used in the following examples, other addresses are possible.

If both ADRS'D and REM still do not light, consult the 8566A/8568A Operating and Service Manual and the 9826A/9836A Service Manual for troubleshooting information.

Programming Examples

The following examples illustrate some of the ways to operate the 8566A/8568A using the 9826A/9836A controller.

The examples illustrate setting front panel controls remotely and outputting their values, outputting marker values, and outputting trace data. An example harmonic distortion measurement program incorporates some of these techniques in a typical application.

EXAMPLE 1: PROGRAMMING FRONT-PANEL FUNCTIONS

To preset the analyzer, and set center frequency to 100 MHz and span to 10 MHz, enter the following on the keyboard of the 9826A/9836A controller: OUTPUT 718; "IP CF100MZ SP10MZ"



Executing this statement initiates the sequence of operations shown above. The final CRT display with a 100 MHz signal present should look like this:



The last function activated, SPAN, will appear with its current value on the analyzer CRT as shown in the shaded box.

NOTE

An important concept in analyzer programming is worthy of special note here. The sequence of operations executed above could have been entered manually from the front panel of the analyzer to yield the same result. In fact, a manual sequence of keystrokes is usually developed first and then used as a basis for executing the same procedure under program control. This simple technique is recommended as a powerful tool for software development with the automatic spectrum analyzer.

EXAMPLE 2: OUTPUTTING A FUNCTION OR MARKER VALUE

In the first case, a BASIC program is shown which directs the analyzer to activate center frequency, and to prepare to output the current value in a subsequent statement. The value is then transferred into the variable F and printed. The END statement, line 50, terminates the program.

```
10    OUTPUT 718;"CF OA"                  ! Activate center frequency,prepare
20                                        ! to output value of active function.
30    ENTER 718;F                         ! Transfer value to F.
40    PRINT "Center Frequency =";F;"Hz"   ! Print value.
50    END
```

To enter the program, press:

EDIT (Press EXECUTE.)

10_

should appear. Type a line and press ENTER. Now

20_

should appear. Continue entering program code line by line. After storing the last line, END, press RUN to execute the program.* (Omit annotation which begins with "!" on each line, or entire lines which contain only annotation; these comments are provided for the reader's clarification only. Note that your line numbers will not in general correspond to those in this guide.)

A typical output would be:

```
Center Frequency = 1.E+8 Hz
```

Next, we would like to output both the amplitude and frequency of the active marker. To illustrate this, connect the analyzer's CAL OUTPUT to the RF INPUT. Type SCRATCH A and press EXECUTE to clear the program memory, and enter the following program:

```
10    OUTPUT 718;"IP FA75MZ FB150MZ S2 TS E1"
20                                        ! Instrument  preset, set start and
30                                        ! stop freq's, single sweep,
40                                        ! take sweep, peak search.
50    OUTPUT 718;"MA"                     ! Prepare to output marker amplitude.
60    ENTER 718;A                         ! Transfer amplitude into variable A.
70    OUTPUT 718;"MF"                     ! Prepare to output marker frequency.
80    ENTER 718;F                         ! Transfer frequency into variable F.
90    PRINT A;"dBm   ";F/1.E+6;"MHz"      ! Print A and F (scaled to megahertz).
100   END
```

The first line presets the analyzer, sets start and stop frequencies to 75 MHz and 150 MHz, and then instructs the analyzer to use the single sweep mode. To ensure that a trace is displayed which corresponds to the current instrument control settings, a take sweep command ("TS") is used. This arms the sweep, causing a sweep to be taken when trigger conditions are met, and prevents the analyzer from accepting further commands until the trace is complete.

*For a brief introduction to the controller editing facilities, refer to the chapter entitled Entering, Running and Storing Programs in BASIC Programming Techniques for the 9826A/9836A.

Upon completion of this sweep, the peak search ("E1") command is invoked, placing a marker on the largest signal displayed. Lines 50 and 60 instruct the analyzer to output the amplitude value in dBm into the variable A, and lines 70 and 80 cause the frequency in hertz to be transferred into F. These two values are then printed with appropriate units. Note that the frequency in hertz has been divided by one million to yield megahertz.

Pressing RUN yields typical output:

```
-10.4 dBm     100.2 MHz
```

EXAMPLE 3: OUTPUT TRACE DATA

An important capability of an automatic spectrum analyzer is to transfer trace amplitude data into an array in the controller for subsequent manipulation. A direct approach is shown in the first program:

```
10    DIM A(1000)                      ! Dimension array A from 0 to 1000.
20                                     ! (1001 points total).
30    OUTPUT 718;"S2 TS O3 TA"         ! Using O3 format (reference level units),
40                                     ! prepare to output trace A.
50    FOR N=0 TO 1000                  ! Begin FOR-NEXT loop.
60    ENTER 718;A(N)                   ! Transfer formatted data one point at a time
70                                     ! into A array..
80    NEXT N                           ! End of FOR-NEXT loop.
90    FOR N=0 TO 1000 STEP 100         !
100   PRINT N,A(N)                     ! Print every one-hundredth point.
110   NEXT N                           !
120   END
```

After dimensioning the array, four commands are sent to the analyzer in the OUTPUT 718 statement. First, the analyzer is set to the single sweep mode, followed by a take sweep command. The single sweep mode ("S2") is especially important when outputting trace data because it provides a static display while the values are being accessed. Following the TS command (discussed in Example 2) there is an output format command O3. (This is the letter O for Output, not zero!) The analyzer in this mode scales the display units from the ADC (analog-to-digital converter) to reference level units (in this example, dBm), and re-formats these values into a sequence of ASCII characters which will be transmitted over the interface bus. TA specifies trace A data, which are subsequently transferred one point at a time into the A array using the ENTER 718 statement repeated 1001 times.

Finally, to show what has happened, several data values are printed.

```
0         -86.5
100       -83.4
200       -83.8
300       -87.4
400       -82.9
500       -13
600       -79.1
700       -83.1
800       -84.5
900       -83.6
1000      -87.5
```

The execution time for the trace data transfer in O3 format using the 9826A is about 3.6 seconds. To achieve a faster transfer, we can avoid rescaling the ADC values and re-formatting into ASCII code by using O2 instead of O3 output format. The trace data can then be transferred as unformatted binary values by using an I/O path with the ASCII format off.

In the case below, a sequence of 8-bit bytes is transferred into the integer-valued A array. Note that the values in the A array are two bytes or sixteen bits long, as are the binary values to be transferred from the analyzer in the O2 format mode. The values printed from the A array are in display units. These range from 0 to 1023, and may be accessed as such for further processing. A typical execution time for this transfer using the 9826A is 150 milliseconds.

```
10    INTEGER A(1000)                    ! Dimension A array from 0 to 1000
20                                       ! (1001 points total).
30    OUTPUT 718;"S2 TS O2 TA"           ! Single sweep, take sweep, using format O2
40                                       ! (binary units) prepare to output trace A.
50    ASSIGN @Sa TO 718;FORMAT OFF       ! Assign I/O path @Sa to spectrum analyzer
60                                       ! and turn ASCII format off for this path.
70    ENTER @Sa;A(*)                     ! Transfer data into array using specified
80                                       ! I/O path with format off.
90    FOR N=0 TO 1000 STEP 100           !
100   PRINT N,A(N)                       ! Print every one-hundredth point.
110   NEXT N                             !
120   END
```

```
0       139
100     178
200     127
300     129
400     169
500     870
600     207
700     154
800     126
900     131
1000    160
```

This program illustrates how more advanced BASIC programming techniques can be implemented to produce significantly higher performance in the area of automatic instrument control. Such topics as advanced transfer techniques are treated in BASIC Interfacing Techniques for the 9826A/9836A.

NOTE

Correct format usage when transferring data and commands to and from the analyzer is essential for proper operation under remote control. Errors in formatting are a frequent cause of program failure; study the format codes if you are not certain of correct usage when debugging a program under development.

Data are transferred over the interface bus one 8-bit byte at a time. These may be ASCII-encoded alphanumeric characters, or binary values. For example, when the O3 format has been specified (this is the default mode on instrument preset) and a trace value is output from the analyzer, a sequence of ASCII characters is transmitted across the bus, as many as needed to specify the value of interest. The analyzer automatically performs the necessary formatting from an internally stored binary value to an ASCII string, and the controller reverses this process on receipt of such a string. As the number of characters transferred is variable, a free field format is required in the control program.

Alternatively, data values themselves may be transferred in 8-bit bytes (two bytes will be necessary to retain the full 10-bit precision of values stored in

the analyzer). Here, the analyzer may be in the O2 format, and the controller in an unformatted or binary formatted mode (i.e., ASCII formatting must not occur). This is illustrated in the second trace output example involving the format off I/O path.

See the Spectrum Analyzer Remote Operation manual for further information on input/output formats.

EXAMPLE 4: HARMONIC DISTORTION MEASUREMENT

An example program which illustrates some of the techniques demonstrated above is included here. This program makes a harmonic distortion measurement by locating and measuring a signal's second and third harmonics and calculating the percent distortion relative to the fundamental. The technique suggested in Example 1 – converting a manual sequence of keystrokes into a program to perform the same functions – was used in developing the present example.

```
10    ! HARMONIC DISTORTION MEASUREMENT
20    !
30    OUTPUT 718;"IP"
40    LOCAL 718
50    DISP "Set analyzer to display the fundamental signal."
60    PAUSE
70    DISP ""
80    OUTPUT 718;"SP O3 OA"               ! Prepare to output the current span.
90    ENTER 718;Span                      ! Transfer value (in hertz) to "Span".
100   IF Span<=1.E+5 THEN 120             ! Use current value or 100 KHz,
110   Span=1.E+5                          ! whichever is smaller.
120   OUTPUT 718;"S2 TS E1 MT1 SP";Span;"HZ TS MT0 E4 TS E1 E3 MA"
130             ! Acquire signal with peak search, auto-zoom, marker to
140             ! reference level, peak search; enter CF STEP SIZE with E3
150             ! command; use MA to prepare to output fundamental amplitude.
160   ENTER 718;Fund                      ! Transfer marker amplitude to "Fund".
170   OUTPUT 718;"MF"                     ! Prepare to output marker frequency.
180   ENTER 718;Freq                      ! Transfer marker freq to "freq".
190   Freq=Freq/1.E+6                     ! Scale frequency to megahertz.
200   OUTPUT 718;"CF UP TS E1 MA"         ! Increment center freq by fundamental freq.
210   ENTER 718;Second                    ! Transfer marker amplitude to "Second".
220   OUTPUT 718;"CF UP TS E1 MA"         ! Increment center freq by fundamental freq.
230   ENTER 718;Third                     ! Transfer marker amplitude to "Third".
240   Dist=100*SQR(FNLin(Second)^2+FNLin(Third)^2)/FNLin(Fund)
250                                       ! Compute root-sum-of-squares
260                                       ! total harmonic distortion using "Lin"
270                                       ! function defined below.
280                                       !
290 Format1:    IMAGE 4A,X,SDDD.D,X,"dBm",3X,K,X,"MHz" !_____
300   PRINT USING Format1;"Fund",Fund,Freq              !_____     Formated
310 Format2:  IMAGE 2(4A,X,SDDD.D,X,"dBm",/)            !_____       output
320   PRINT USING Format2;"2nd ",Second,"3rd",Third     !_____
330   PRINT USING "K,DDD.DD,K,//";"Harmonic Distortion = ";Dist;"%"
340   END
350   !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
360   DEF FNLin(X)                  !
370   Lin_value=10^(X/20)           ! Function to compute linear value from dB's.
380   RETURN Lin_value              !
390   FNEND                         !
```

Line 40 places the analyzer under front panel control allowing the operator to tune the analyzer to position the signal on screen. The span must be chosen such that the signal of interest is the largest response on the screen. When ready, the operator presses CONTINUE. The program determines the present span and compares it to 100 kHz, choosing the smaller value. Then, a sweep is taken in single sweep mode, and peak search places the marker on the largest signal, i.e., the fundamental. Marker track is invoked to perform an Auto-Zoom to the span selected above. The signal is then moved to the reference level, the center frequency step size is set to the fundamental frequency, and the amplitude and frequency are output to the controller.

In line 200, the center frequency is incremented once to place the second harmonic on screen. Peak search locates the response and the marker amplitude is output. The same procedure is performed on the third harmonic in line 220.

In line 240, the percent distortion is computed as the root sum of the squares normalized to the fundamental amplitude. As linear values are required in this calculation, a function has been defined in lines 360 – 390 which converts the dBm values to linear values. The results are finally printed according to the output formats in lines 290 – 330*.

A typical harmonic distortion measurement might yield the following output:

```
Fund   -10.3 dBm    20.0005 MHz
2nd    -55.5 dBm
3rd    -67.3 dBm

Harmonic Distortion =     .57%
```

FOR MORE TRAINING

50003A
8566A/8568A Spectrum Analyzer
Operation Course

The 50003A Spectrum Analyzer Operation Course provides comprehensive training in the remote operation of the 8566A and 8568A. This intensive four-day course teaches manual and remote operating techniques and signal measurement concepts as they apply to these analyzers. The curriculum is heavily hands-on oriented, using a mixture of interactive lectures and labs with the 9826A as instrument controller.

The 50003A is offered at selected HP training centers. Please contact your local HP sales office for scheduling and price information.

*A discussion of PRINT and IMAGE statements can be found in BASIC Programming Techniques for the 9826A/9836A.

8568A PROGRAMMING CODE LIST

FRONT PANEL COMMANDS

AT	Input attenuation
* A1	Clear-write trace A
A2	Max Hold trace A
A3	Store and view trace A
A4	Store and blank trace A
BL	B − DL → B
B1	Clear-write trace B
B2	Max hold trace B
B3	Store and view trace B
* B4	Store and blank trace B
* CA	Coupled input attenuation
CF	Center frequency
* CR	Coupled resolution BW
* CS	Coupled step size
* CT	Coupled sweep time
* CV	Coupled video BW
* C1	A − B off
C2	A − B → A
DB	dB
DL	Display line
DM	dBm
DN	Step down
DT	Label terminator
EE	Enable number entry
EK	Enable DATA knob
* EM	Erase trace C memory
EX	Exchange A and B
E1	Peak Search
E2	Enter marker into center frequency
E3	Enter marker/Δ frequency → step size
E4	Enter marker amplitude → reference level
* FA	Start frequency
* FB	Stop frequency
FS	0 - 1.5 GHz span
GZ	GHz
* HD	Hold
HZ	Hz
IP	Instrument preset
I1	Left RF input
* I2	Right RF input
KS	Shift front panel keys
* KSA	Amplitude in dBm
KSB	Amplitude in dBmV
KSC	Amplitude in dBuV
KSD	Amplitude in voltage
KSE	Title
KSF	Measure sweep time
KSG	Video averaging on
* KSH	Video averaging off
KSI	Extended reference level range
KSJ	Manual DAC control
KSK	Count pilot IF at marker
KSL	Noise level off
* KSM	Noise level on
KSN	Count VTO at marker
KSO	Enter Δ → span
KSP	Set HP-IB address
KSQ	Count signal IF
KSR	Diagnostics on
* KSS	Second LO auto
KST	Second LO down
KSU	Second LO up
KSV	Frequency offset
KSW	Error correction routine
KSX	Use correction data
KSY	Do not use correction data
KSZ	Amplitude offset
* KSa	Normal detection
KSb	Positive peak detection
KSc	A + B → A
KSd	Negative peak detection
KSe	Sample detection
KSf	Power on in last state
KSg	CRT beam off
KSh	CRT beam on
KSi	Exchange B and C
KSj	View trace C
KSk	Blank trace C
KSl	Trace B → trace C
KSm	Graticule blanked
* KSn	Graticule on
KSo	Characters blanked
* KSp	Characters on
KSq	Step gain off
KSr	Service request 102
KSt	Continue sweep from marker
KSu	Stop at marker, single sweep
KSv	Inhibit phase lock
KSw	Display correction data
KSx	normal EXT trigger
KSy	normal VID trigger
KSz	Display storage address
KS,	Mixer level
KS −	Negative entry
KS =	Counter resolution
KS(	Save registers locked
KS)	Save registers unlocked
KS >	Preamp gain, input 2
KS <	Preamp gain, input 1
KS|	Display storage write
KZ	kHz
* LG	Enter log scale
LN	Linear scale
L0	Display line off
* MC0	Marker frequency count off
MC1	Marker frequency count on
MS	msec
MV	mV
* MT0	Marker signal track off
MT1	Marker signal track on
MZ	MHz
* M1	Marker off
M2	Marker normal
M3	Marker Δ
M4	Marker zoom
RB	Resolution BW
RC	Recall
RL	Reference level
SC	sec
SP	Frequency span
SS	Center frequency step size
ST	Sweep time
SV	Save
* S1	Sweep continuous
S2	Sweep single
TH	Enter threshold
* T0	Threshold off
* T1	Trigger free run
T2	Trigger line
T3	Trigger external
T4	Trigger video
UP	Step up
UR	Upper right
US	μsec
UV	μV
VB	Video BW
0 to 9	0 to 9
.	Decimal point or period

OUTPUT COMMANDS

DR	Read display and increment address
EE	Enable number entry
KS123$_{10}$	Output up to 1001 words
LL	Lower left recorder output
MA	Marker amplitude output
MF	Marker frequency output
OA	Output active function
OL	Output learn string
OT	Output display text
O1	Output format ASCII display units
O2	Output format two 8 bit binary bytes
* O3	Output format ASCII parameter or instrument units
O4	Output format one 8 bit binary byte
TA	Output trace A
TB	Output trace B
UR	Upper right recorder output

DISPLAY INPUT COMMANDS

* DA	Display address
DD	Display write
DW	Write into display and increment address
* D1	Display size normal
D2	Display size full CRT
D3	Display size expand
GR	Graph
IB	Input trace B, binary
KS125$_{10}$	Input up to 1001 display memory words
LB	Label
PA	Plot absolute
* PD	Pen down
PR	Plot relative
PS	Skip to next display page
PU	Pen up
SW	Skip to next control instruction
TS	Take sweep

SERVICE REQUEST COMMANDS

R1	Allow only SRQ 140
R2	Allow SRQ 140 and 104
* R3	Allow SRQ 140 and 110
R4	Allow SRQ 140 and 102

SRQ	Command	Bit	Definition
102	R4	1	units key pressed
104	R2	2	end of sweep
110	R3	3	hardware broken
140	all	5	illegal command
1xx	—	6	universal HP-IB service

* selected with instrument preset

8566A PROGRAMMING CODE LIST

FRONT PANEL COMMANDS

AT	Input attenuation
* A1	Clear-write trace A
A2	Max Hold trace A
A3	Store and view trace A
A4	Store and blank trace A
BL	B − DL → B
B1	Clear-write trace B
B2	Max hold trace B
B3	Store and view trace B
* B4	Store and blank trace B
* CA	Coupled input attenuation
CF	Center frequency
* CR	Coupled resolution BW
* CS	Coupled step size
* CT	Coupled sweep time
* CV	Coupled video BW
* C1	A − B off
C2	A − B → A
DB	dB
DL	Display line
DM	dBm
DN	Step down
DT	Label terminator
EE	Enable number entry
EK	Enable DATA knob
* EM	Erase trace C memory
EX	Exchange A and B
E1	Peak Search
E2	Enter marker into center frequency
E3	Enter marker/Δ frequency → step size
E4	Enter marker amplitude → reference level
* FA	Start frequency
* FB	Stop frequency
GZ	GHz
* HD	Hold
HZ	Hz
IP	Instrument preset
KS	Shift front panel keys
* KSA	Amplitude in dBm
KSB	Amplitude in dBmV
KSC	Amplitude in dBuV
KSD	Amplitude in voltage
KSE	Title
KSF	Measure sweep time
KSG	Video averaging on
* KSH	Video averaging off
KSI	Extended reference level range
KSJ	DAC control
KSK	Marker to next peak
KSL	Noise level off
* KSM	Noise level on
KSN	Marker to minimum
KSO	Enter Δ → span
KSP	Set HP-IB address
KSQ	Band unlock
KSR	Diagnostics on
KSS	Fast HP-IB
KST	Fast preset 2 — 22 GHz
KSU	External mixer preset
KSV	Frequency offset
KSW	Error correction routine
KSX	Use correction data
KSY	Do not use correction data
KSZ	Amplitude offset
* KSa	Normal detection
KSb	Positive peak detection
KSc	A + B → A
KSd	Negative peak detection
KSe	Sample detection
KSf	Power on in last state
KSg	CRT beam off
KSh	CRT beam on
KSi	Exchange B and C
KSj	View trace C
KSk	Blank trace C
KSl	Trace B → trace C
KSm	Graticule blanked
* KSn	Graticule on
KSo	Characters blanked
* KSp	Characters on
KSq	Step gain off
KSr	Service request 102
KSt	Band lock
KSu	Stop at marker, single sweep
KSv	Signal identifier ext mixer
KSw	Display correction data
KSx	normal EXT trigger
KSy	normal VID trigger
KSz	Display storage address
KS,	Mixer level
KS=	Factory preselector setting
KS −	Negative entry
KS(	Save registers locked
KS)	Save registers unlocked
KS;	Display storage write
KS#	Turns off YTX self-heating correction
KS/	Manual Preselector peak
KS<9 2>	Enter DL, TH, M2, M3 in display units
KZ	kHz
LF	Preset 0 — 2.5 GHz
* LG	Enter log scale
LL	Lower Left
LN	Linear scale
LO	Display line off
MS	msec
MV	mV
* MT0	Marker signal track off
MT1	Marker signal track on
MZ	MHz
* M1	Marker off
M2	Marker normal
M3	Marker Δ
M4	Marker zoom
PP	Preselector peak
RB	Resolution BW
RC	Recall
RL	Reference level
SC	sec
SP	Frequency span
SS	CF step size
ST	Sweep time
SV	Save
* S1	Sweep continuous
S2	Sweep single
TH	Enter threshold
* T0	Threshold off
* T1	Trigger free run
T2	Trigger line
T3	Trigger external
T4	Trigger video
UP	Step up
UR	Upper right
US	μsec
UV	μV
VB	Video BW
0 to 9	0 to 9
.	Decimal point or period

OUTPUT COMMANDS

DR	Read display and increment address
EE	Enable number entry
KS<91>	Output amplitude error
KS<94>	Output code for harmonic number in binary
KS<123>	Display read binary
KS<126>	Output every nth value of trace
LL	Lower left recorder output
MA	Marker amplitude output
MF	Marker frequency output
OA	Output active function
OL	Output learn string
OT	Output display text
O1	Output format ASCII display units
O2	Output format two 8 bit binary bytes
* O3	Output format ASCII parameter or instrument units
O4	Output format one 8 bit binary byte
TA	Output trace A
TB	Output trace B
UR	Upper right recorder output

DISPLAY INPUT COMMANDS

* DA	Display address
DD	Display write
DW	Write into display and increment address
* D1	Display size normal
D2	Display size full CRT
D3	Display size expand
GR	Graph
IB	Input trace B, binary
KS<39>	Fast Binary DA . . DW
KS<125>	Display write binary
KS<127>	Display write binary
LB	Label
PA	Plot absolute
* PD	Pen down
PR	Plot relative
PS	Skip to next display page
PU	Pen up
SW	Skip to next control instruction
TS	Take sweep

SERVICE REQUEST COMMANDS

R1	Allow only SRQ 140
R2	Allow SRQ 140 and 104
* R3	Allow SRQ 140 and 110
R4	Allow SRQ 140 and 102
KS<43>	Allow SRQ 140 and 102

SRQ	Command	Bit	Definition
102	R4	1	units key pressed
102	KS<43>	1	frequency limit exceeded
104	R2	2	end of sweep
110	R3	3	hardware broken
140	all	5	illegal command
1xx	—	6	universal HP-IB service

* selected with instrument preset

For more information, call your local HP Sales Office or nearest Regional Office: **Eastern** (201) 265-5000; **Midwestern** (312) 255-9800; **Southern** (404) 955-1500; **Western** (213) 970-7500; **Canadian** (416) 678-9430. Ask the operator for instrument sales. Or write Hewlett-Packard, 1501 Page Mill Road, Palo Alto, CA 94304. **In Europe:** Hewlett-Packard S.A., 7, rue du Bois-du-Lan, P.O. Box, CH 1217 Meyrin 2, Geneva, Switzerland. **In Japan:** Yokogawa-Hewlett-Packard Ltd., 29-21, Takaido-Higashi 3-chome, Suginami-ku, Tokyo 168.

5952-9373 Printed in U.S.A.